

高等学校教学参考书



BASIC

在电工电子学中的应用

浩强 周朝龙 编 · 高等教育出版社

内 容 简 介

本书介绍了BASIC语言在电工、电子学中的应用,它包括电路、电子线路、电磁学、电子学、统计方法等方面的典型问题,精选了86个例题,大多数程序都给出了图形结果,以便读者更好地理解电工、电子学中的概念。本书可以作为电类专业学生学习 BASIC 语言课程时的参考书,或作为学完 BASIC 语言后应用于电工、电子学领域时参考。本书也可用于电学课程的计算机辅助教学。

BASIC 在电工电子学中的应用

谭浩强 周朝龙 编

*

高等教育出版社出版
新华书店北京发行所发行
北京印刷一厂印装

*

开本 787×1092 1/16 印张 20.25 字数 500 000

1989年12月第1版 1989年12月第1次印刷

印数 0001—3 330

ISBN7-04-002101-3/TP·48

定价5.00元

前 言

随着计算机应用愈来愈深入各个领域,许多学过计算机语言的同志迫切希望能够运用计算机来解决本领域中的实际问题。

为了帮助学过计算机语言的读者能较快地将计算机应用于各专业领域,全国高等学校计算机基础研究会和全国非计算机专业计算机课程讨论会经过多次研究,提出编写一套有关计算机语言在各领域应用的题选的参考书。例如在电工学、机械、管理、建筑、数理等方面的应用,以供从事各专业领域学习和工作的同志参考。

本书就是为适应这种要求而编写的,作为这类参考书的第一本,同时也作为一个尝试,以抛砖引玉。

由于 BASIC 语言是初学者广泛使用的语言,有很强的会话功能和作图功能,而且在微机上使用很方便,所以本书采取 BASIC 语言来编写程序。

本书的内容是 BASIC 语言在电学方面应用的题选,共包含 86 个例题。主要是电工、电子学中的一些基本问题。每类问题都给出了问题分析、数学方法、BASIC 程序以及运行结果。在运行结果中,除了给出计算的结果数据外,大多数例题还同时给出了图形结果,使结果更加清晰、直观。这些典型的例题,可以作为解决一些复杂问题的基础。

此外,本书还可以用于计算机辅助教学。在电学中许多复杂的概念,象电和磁的效应是肉眼所看不见的,这就造成学习中的一定困难。利用计算机将电学中的计算公式转换成程序,只要送入不同的参数,就可以得到不同的结果,并且用图形表示出来。从而改变了传统的电学实验方式。用计算机来进行电学的实验,手段更为先进,速度更快,结果更直观,而且还能解决许多传统实验所无法解决的问题。计算机技术和各个学科相结合,将推动各个学科的深入发展。

本书例题选择的原则是:尽量选择那些能充分体现计算机作用的一些问题(例如,为复杂的问题提供解题程序,用图形显示结果以帮助理解等)。凡是通过简单的公式计算就能直接得到计算结果的,我们一般不予选入。

本书包括电路、电子线路、电磁学、电子学、统计方法等五章。在本书附录中,提供了在处理电学问题时所常遇到的数值计算方法和程序,以减少读者在阅读本书中的困难。

在各例题的运行结果中,有打印出的该程序运行时所花费的时间(个别程序的运行时间比较长),以供使用时参考。

读者在使用程序时,自己改变程序中一些参数,就可以得到不同的结果和图形,读者可以对它们进行比较分析,从而加深对所学电学知识的理解。

在编写本书时,参考了日本须由健二、北原记之二人著的《BASIC 的电气、电子》一书。西北轻工业学院狄剑忠同志参加了部分外文资料的翻译工作。李林森同志校阅了本书部分章节,王玉如同志参加了有关插图工作。没有他们的参加,本书是难以完成的。本书出版前,承清华大学江缉先、胡蓉芳两同志审阅了全部书稿,提出不少宝贵意见,谨此一并表示感谢。

为了方便读者使用本书程序,我们已将本书的全部程序存入磁盘,需要者可与高等教育出版社联系购买。磁盘中除了本书全部程序外,还有 BASICA 和 GRAPHICS (屏幕拷贝图形软

件)。如果需要打印图形,应在 DOS 调入内存后,先运行 GRAPHICS 然后再调入 BASICA,用 LOAD“文件名”命令调用所需的程序并运行。为了读者查用程序方便,本书提供了一个程序索引,指出程序文件名、程序内容以及在本书的页数。

编写本书的目的是为了推动计算机在各个领域中的应用。由于经验不足,本书定有许多不妥之处,敬请读者批评指正,同时我们期望有更多的这类参考书的出现,以满足社会各方面的要求。

编 者

1988.2

目 录

使用本书的注意事项

第一章 电路 1

§ 1.1 交流电压,交流电流 1

1.1.1 交流电压、电流的平均值和有效值 1

1.1.2 交流电压·电流·功率 3

1.1.3 矢量图 6

§ 1.2 交流电路和矢量轨迹 8

1.2.1 RL 串联电路的矢量轨迹 8

1.2.2 低通滤波电路的矢量轨迹 11

1.2.3 低通滤波电路的频率特性 13

§ 1.3 一般线性网络 15

1.3.1 直流电路的分析 16

1.3.2 交流电路的分析 19

§ 1.4 非正弦周期信号 23

1.4.1 傅立叶级数展开 23

1.4.2 离散的傅立叶分析 27

1.4.3 非正弦周期信号的谐波分析 33

§ 1.5 三相交流电路 37

1.5.1 不平衡三相电路的分析 37

§ 1.6 过渡过程 40

1.6.1 $RLC-RLC$ 电路过渡过程的数值解法 40

1.6.2 $RLC-RLC$ 电路过渡过程的数式*解法 47

§ 1.7 分布参数电路 62

1.7.1 正弦交流电压时的分布参数电路 62

习题 66

第二章 电子线路 67

§ 2.1 小信号放大电路 67

2.1.1 RC 耦合放大电路 67

§ 2.2 功率放大电路 72

2.2.1 输出波形的谐波分析和波形失真因数 73

§ 2.3 负反馈放大电路 75

2.3.1 双T网络的频率特性 76

2.3.2 选频放大电路的频率特性 80

§ 2.4 振荡电路 82

2.4.1 调谐式振荡电路 82

2.4.2 文氏电桥振荡电路 88

§ 2.5 调制与解调电路 92

2.5.1 AM波,BM波*,FM波 92

2.5.2 贝塞尔函数的数值计算 98

§ 2.6 电源电路 102

2.6.1 全波整流和滤波电路 102

§ 2.7 脉冲电路 105

2.7.1 RC 串联回路的脉冲响应 105

习题 109

第三章 电磁学 110

§ 3.1 电场和电位 110

3.1.1 库仑定律 110

3.1.2 电场和电力线 110

3.1.3 电位和等位面 111

3.1.4 电偶极子 112

§ 3.2 导体系 122

3.2.1 电位系数 122

3.2.2 电容系数和感应系数 126

§ 3.3 静电镜像法和保角映射 130

3.3.1 静电镜像法 130

3.3.2 保角映射 139

§ 3.4 泊松方程和拉普拉斯方程 150

3.4.1 泊松方程 150

3.4.2 拉普拉斯方程 156

§ 3.5 电流和磁场 169

3.5.1 安培右手螺旋定律 169

3.5.2 毕奥-萨伐定律 170

3.5.3 弗莱明左手定律 175

§ 3.6 电磁感应 181

3.6.1 电感.....	181	5.1.4 正态分布及其随机数	236
3.6.2 诺埃曼公式	181	5.1.5 二元正态分布和二元正态 随机变量	240
习题	188	§5.2 时间序列分析	244
第四章 电子学	189	5.2.1 随机变量的变换	245
§4.1 电磁场中电子的运动	189	5.2.2 移动平均法	249
4.1.1 电场中电子的运动	189	5.2.3 相关函数	253
4.1.2 非均匀电场中电子的运动	192	5.2.4 快速傅立叶变换	267
4.1.3 均匀磁场中的电子运动	195	5.2.5 最小二乘法	275
4.1.4 电磁场中的电子运动	197	习题	284
§4.2 半导体的物理现象	201	附录	285
4.2.1 电子能级分布	201	§1 二重积分的计算	285
4.2.2 半导体的电导	206	§2 应用贝尔斯特法解高次代数 方程的复数根	287
4.2.3 隧道效应	209	§3 应用基尔法解常微分方程.....	295
§4.3 脉冲信号	214	§4 联立常微分方程组的求解 ...	298
4.3.1 脉冲波形及脉冲变换	214	§5 高阶常微分方程的求解	299
4.3.2 微分电路	218	§6 偏微分方程的求解	301
4.3.3 积分电路	223	(1)椭圆型偏微分方程的求解	302
习题	224	(2)抛物型偏微分方程的求解	306
第五章 统计方法	225	(3)双曲型偏微分方程的求解	307
§5.1 随机数的产生及其分布 ...	225	例题索引	309
5.1.1 均匀分布及均匀随机数	225	参考文献	312
5.1.2 二项分布及二项随机数	228		
5.1.3 泊松分布及泊松随机数	231		

第一章 电 路

通过本章的学习,可以基本掌握用微机来解决电路问题的一般技巧.

由于电路的涉及面很广,在此仅举一些有代表性的例子来说明用微机解决这类问题的方法,并尽可能地用绘图方法来加深理解.

另外,在解电路题时,常常会碰到复数问题.因此,我们也列举了不少用 BASIC 来处理复数问题的例题.

§ 1.1 交流电压,交流电流

我们称方向随时间周期性变化而在一个周期内其平均值为零的电压和电流为**交流电压**和**交流电流**.交流电具有各种各样的波形,而其最基本的波形是如图 1-1 所示的正弦波,它的数学表示式为:

$$v(t) = E_m \sin(\omega t - \theta) \quad (1.1)$$

这里的 $v(t)$ 称为电压的**瞬时值**, E_m 称为**最大值**, ω 是**角频率**,它和**周期** T 之间具有以下关系:

$$\omega T = 2\pi \quad (1.2)$$

因为频率 f (赫兹) 是 $1/T$, 所以,

$$\omega = \frac{2\pi}{T} = 2\pi f \quad (1.3)$$




图 1-1 正弦交流电压

式(1.1)中的 θ 是**初相位角**,用来确定当 $t=0$ 时的波形位置.

1.1.1 交流电压、电流的平均值和有效值

仅用最大值来表示交流电压或电流的大小并不能代表交流电压或电流的实际大小,为此,还定义了交流电压或电流的**平均值**和**有效值**.

所谓交流电压(或电流) $v(t)$ 的平均值是指将负半周的波形反转,再求出在一个周期中的平均值.可以用下式来定义:

$$V_{\text{ave}} = \frac{1}{T} \int_0^T |v(t)| dt \quad (1.4)$$

所以,对于象正弦波这样一种正负对称的波形,可以用正半波来求平均值.我们把**瞬时值的均方根**称为**有效值**,并用 $|V|$ 来表示,定义为:

$$|V| = \sqrt{\frac{1}{T} \int_0^T v^2(t) dt} \quad (1.5)$$

交流电除了大小这一因素之外,有时还必须考虑其波形的形状.我们用**波形因数**来表示波

形的光滑度,用**峰值因数**来表示波形的锐度,它们分别用以下两式定义:

$$\text{波形因数} = \frac{\text{有效值}}{\text{平均值}} \quad (1.6)$$

$$\text{峰值因数} = \frac{\text{最大值}}{\text{有效值}} \quad (1.7)$$

例题 1 (DIANLU.001) 利用数值积分计算正弦交流波的平均值、有效值,并求波形因数及峰值因数。请编写该程序。

设正弦交流波为:

$$v(t) = E_m \sin \omega t$$

按下式可求得平均值 V_{ave} (程序中以 Ave 表示)和有效值 $|V|$ (程序中以 Rms 表示):

$$V_{\text{ave}} = \frac{2}{T} \int_0^{\frac{T}{2}} E_m \sin \omega t dt = \frac{2}{\pi} E_m \doteq 0.636 E_m \quad (1.8)$$

$$|V| = \sqrt{\frac{1}{T} \int_0^T (E_m \sin \omega t)^2 dt} = \frac{E_m}{\sqrt{2}} \doteq 0.707 E_m \quad (1.9)$$

于是,就可按以下两式求得波形因数(FF.)及峰值因数(CF.):

$$\text{波形因数} = \frac{\frac{E_m}{\sqrt{2}}}{\frac{2}{\pi} E_m} = \frac{\pi}{2\sqrt{2}} \doteq 1.111 \quad (1.10)$$

$$\text{峰值因数} = \frac{\frac{E_m}{\sqrt{2}}}{\frac{E_m}{\sqrt{2}}} = \sqrt{2} \doteq 1.414 \quad (1.11)$$

下面给出程序(DIANLU.001)及其运行结果。程序中设 $v(t) = 100 \sin 100\pi t$, 数值积分中(1230 ~ 1320)使用的是**辛普生(Simpson)**求积公式。程序中的 T 为周期, $T = \frac{1}{f} = \frac{1}{50} = 0.02$ 秒, N 为划分小积分区域的个数。

```
1000 'DIANLU.001 (Ave, Rms, FF, CF.)
1010 TIME$ = "00:00:00 "
1020 READ EM, T, N
1030 DATA 100, 0.02, 30
1040 PI=3.14159
1050 DEF FNF1 (X)=EM*SIN(2*PI*X/T)
1060 DEF FNF2 (X)=FNF1 (X)
1070 LPRINT "** sine wave **"
```

```

1080 H=T/2/N/2
1090 GOSUB 1230
1100 AVE=S/.01
1110 LPRINT USING "Ave= ###.##### "; AVE
1120 DEF FNF2 (X)=(FNF1 (X))^2
1130 H=2*H
1140 GOSUB 1230
1150 RMS=SQR (S/.02)
1160 LPRINT USING "Rms= ###.##### "; RMS
1170 LPRINT USING "Max= ###.##### "; EM
1180 FF=RMS/AVE : CF=EM/RMS
1190 LPRINT USING "FF= ###.##### "; FF
1200 LPRINT USING "CF= ###.##### "; CF
1210 LPRINT : LPRINT "time="; TIME $
1220 END
1230 'subroutind Simpson
1240 S=0
1250 FOR K=1 TO N
1260 XA=C+2*(K-1)*H
1270 XB=XA+H
1280 XC=XB+H
1290 S=S+(FNF2 (XA)+FNF2 (XC)+4*FNF2 (XB))
1300 NEXT K
1310 S=S*H/3
1320 RETURN

```

运行结果

```

** sine wave **
Ave = 63.66204
Rms= 70.71071
Max=100.00000
FF = 1.11072
CF = 1.41421
time=00 : 00 : 09

```

1.1.2 交流电压·电流·功率

如图 1-2 所示,在电阻 R 和电感 L 串联电路的两端,加上正弦交流电压:

$$e = E_m \sin (\omega t - \theta)$$

在稳定状态下,电流的瞬时值 i 为:

$$i = I_m \sin (\omega t - \theta - \varphi)$$

式中




图 1-2 $R-L$ 串联电路

而瞬时电功率为:

$$I_m = \frac{E_m}{\sqrt{R^2 + (\omega L)^2}}, \quad \varphi = \operatorname{tg}^{-1} \frac{\omega L}{R}$$

$$p = ei = \frac{1}{2} E_m I_m \{ \cos \varphi - \cos (2\omega t - 2\theta - \varphi) \} \quad (1.13)$$

所以平均电功率为:

$$\begin{aligned} P &= \frac{2}{T} \int_0^{\frac{T}{2}} p dt = \frac{1}{2} E_m I_m \cos \varphi \\ &= |E| |I| \cos \varphi \quad [\text{W—瓦特}] \end{aligned} \quad (1.14)$$

例题 2 (DIANLU.002) 当在图1-2中的 RL 串联电路上加上正弦交流电时, 试编程序来求电路中电流的瞬时值、瞬时功率, 并将其同时间的关系用图表示出来。

下面是程序 DIANLU.002, 图1-3表示其执行结果。程序中设 $e = 100\sqrt{2} \sin(100\pi t + 0^\circ)$, $R = 10 \Omega$, $L = 0.01\text{H}$ 。程序变量和公式中对应定义符的关系为: $EA = 100$, 所以 $E_m = EM = 100\sqrt{2} = \sqrt{2} EA$ (1160行); $2\pi f = 100\pi$, 所以 $f = FR = 50$ (1040行); $\omega = 2\pi f$, 即 $W = 2 * PI * FR$ (1130行); $\theta' = DO$ (度) $= 0^\circ$, 所以 $-\theta = RA1 = PI * DO / 180$ (弧度) (1130行); $\varphi = -RA2 = -\operatorname{tg}^{-1} \frac{\omega L}{R}$ (1140行); $Z = \sqrt{R^2 + (\omega L)^2}$ (1150行), 所以 $I_m = IM = EM / Z$ (1170行)。另外, G, Q, A, S 分别为 t, e, i, p 的倍率; 瞬时值 t, e, i, p 分别对应程序变量 T, E, I, P ; EA, IA, PA 分别为电压、电流、功率的有效值程序变量。

R 和 L 的值可以在 1070 行上改变。请你改变后试着执行一下看看, 这时可能会因为 e, i 和 p 值的过大或过小而使图象看不到。在 1080 ~ 1110 语句中适当调整其轴的倍率, 就可使图象固定在屏幕的合适位置上。

本例中输入数据为: $T = 10000, E = 0.5, I = 3, P = 0.08$ 。

```

1000 REM DIANLU.002
1010 TIME $ = "00 : 00 : 00 "
1020 COLOR 8, 0
1030 SCREEN 1, 0
1040 READ EA, FR, DO
1050 DATA 80, 90, 0
1060 READ R, L
1070 DATA 10, 0.01
1080 INPUT "times—ratio of T—axle ", G
1090 INPUT "times—ratio of E—axle ", Q
1100 INPUT "times—ratio of I—axle ", A
1110 INPUT "times—ratio of P—axle ", S
1120 PI = 3.14159

```

```

1130 W=2*PI*FR:RA1=PI*DO/180
1140 RA2=-ATN(W*L/R)
1150 Z=SQR(R*R+(W*L)^2)
1160 EM=SQR(2)*EA:TA=1/FR
1170 IM=EM/Z:IA=IM/SQR(2)
1180 PA=EA*IA*COS(RA2)
1190 X0=10:Y0=100:CLS
1200 LINE(X0,Y0-90)-STEP(0,170)
1210 LINE(X0,Y0)-STEP(390,0)
1220 DT=.0001:T=0
1230 E=EM*SIN(W*T+RA1)
1240 I=IM*SIN(W*T+RA1+RA2)
1250 P=E*I
1260 X1=T*G+X0
1270 PSET(X1,-E*Q+Y0),2
1280 PSET(X1,-I*A+Y0),1
1290 PSET(X1,-P*S+Y0)
1300 T=T+DT
1310 IF T*G>285 THEN 1330 ELSE 1230
1320 '-----
1330 LINE(250,132)-STEP(10,0),2
1340 LINE(250,140)-STEP(10,0),1
1350 LINE(250,148)-STEP(10,0)
1360 LOCATE 1,1:PRINT "e,i,p"
1370 LOCATE 14,35:PRINT "t(sec)"
1380 LOCATE 17,34:PRINT "e"
1390 LOCATE 18,34:PRINT "i"
1400 LOCATE 19,34:PRINT "p"
1410 LOCATE 13,1:PRINT "0"
1420 LOCATE 20,14
1430 PRINT USING "T=####.##(S)";TA
1440 LOCATE 21,14
1450 PRINT USING "E=####.##(V)";EA
1460 LOCATE 22,14
1470 PRINT USING "I=####.##(A)";IA
1480 LOCATE 23,14
1490 PRINT USING "P=####.##(W)";PA
1500 LOCATE 1,27:PRINT "time=";TIME$
1510 LOCATE 23,1
1520 END
RUN
times-ratio of T-axle 10000

```

times—ratio of E—axle 0.5
times—ratio of I—axle 3
times—ratio of P—axle 0.08




图 1-3

1.1.3 矢 量 图

下面我们来研究利用复数将电压、电流作符号化处理,也就是用复数

$$\left. \begin{aligned} E &= \sqrt{2} |E| e^{j(\omega t - \theta)} \\ I &= \sqrt{2} |I| e^{j(\omega t - \theta - \varphi)} \end{aligned} \right\} \quad (1.15)$$

的虚部来表示正弦波电压和电流的瞬时值。式(1.15)表示了按角频率 ω 作反时针方向转动的矢量(它们称为旋转矢量)。E和I无论在什么时刻,两者的相对关系应是不变的。于是可以以 $t=0$ 时的瞬时矢量作为固定矢量(也称相量)来考虑,也就是将 $\dot{E} = |E| \cdot e^{-j(\theta + \varphi)}$ 和 $\dot{I} = |I| \cdot e^{-j(\theta + \varphi)}$ 符号化,这就是图1-4所示的矢量图。由于这些矢量图均按 $t=0$ 的瞬时来选择,故相互间的相位差并不变化。我们把某一矢量恰好和实轴正向一致时的矢量称为基准矢量。




图 1-4 矢量图

例题 3 (DIANLU.003) 在电阻R和电感L所组成的串联电路上加上电源电压 $\dot{E}$, 编程序来计算电路中的电流 $\dot{I}$, 并绘出 $\dot{E}$ 和 $\dot{I}$ 的矢量图。

注: 正弦交流电压和交流电流的瞬时值可以用(1.15)式所示的复数来表示,即可以用复平面内的旋转矢量来表示[由(1.15)式求正弦交流电压和正弦交流电流的瞬时值时,需要对(1.15)式进行反符号化处理,即取其虚部去掉j]。电工学中把 $t=0$ 时的矢量称为相量, $|E|e^{-j\theta}$ 和 $|I|e^{-j(\theta + \varphi)}$ 所表示的矢量称有效值相量(简称相量),相应的图形称为相量图。本节所讲的实际上都是相量图。

图 1-5 表示了程序 DIANLU.003 的执行结果。程序中使用的常数及有关程序变量和上例 (DIANLU.002) 相同, 其中 1090 和 1100 语句是 E 和 I 的倍率输入语句, 目的是为了容易看清所绘制的图形。在实例中, I 被放大了 10 倍, 即实际 I 值是绘制矢量的 $1/10$ 。

本例输入数据为: $E=0.7, I=7$ 。

```

1000 REM DIANLU.003 (vector)
1010 TIME$= "00:00:00"
1020 KEY OFF
1030 SCREEN 1,0:CLS:COLOR 8,0
1040 READ EA,FR,DO
1050 DATA 100,50,0
1060 READ R,L
1070 DATA 10,0.01
1090 INPUT "times-ratio of E is ",A
1100 INPUT "times-ratio of I is ",B
1110 PI=3.14159:W=2*PI*FR
1120 Z=SQR(R^2+(W*L)^2)
1130 RA1=PI*DO/180:RA2=-ATN(W*L/R)
1140 DO2=180*RA2/PI
1150 EM=SQR(2)*EA:IM=EM/Z
1160 PRINT "e=";EM;"SIN(";W;"T+";DO;"")"
1170 PRINT "i=";IM;
1180 PRINT "SIN(";W;"T+";DO+DO2;"")"
1185 LOCATE 11,4:PRINT "Vector indicate";
1190 LINE(300,51)-STEP(-30,0),2
1200 LINE(300,51)-STEP(-4,-4),2
1210 LINE(300,51)-STEP(-4,4),2
1220 LOCATE 7,39:PRINT "E"
1230 LINE(300,67)-STEP(-30,0),1
1240 LINE(300,67)-STEP(-4,-4),1
1250 LINE(300,67)-STEP(-4,4),1
1260 LOCATE 9,39:PRINT "I"
1270 LINE(20,100)-STEP(280,0)
1280 LINE(160,35)-STEP(0,135)
1290 EX=EM*COS(RA1):EY=EM*SIN(RA1)
1300 IX=IM*COS(RA1+RA2)
1310 IY=IM*SIN(RA1+RA2)
1320 XA=EX*A:YA=EY*A:XB=IX*B:YB=IY*B
1330 LINE(160,100)-STEP(XA,-YA),2
1340 LINE(160,100)-STEP(XB,-YB),2
1350 LINE(XA+160,-YA+100)-STEP(-4,-4),2
1360 LINE(XA+160,-YA+100)-STEP(-4,4),2

```

```

1370 LINE(XB+160, -YB+100)-STEP (-4, -4), 1
1380 LINE (XB+160, -YB+100)-STEP (-4, 4), 1
1390 LOCATE 23, 14 : PRINT "time="; TIMES;
1400 LOCATE 22, 1 : END

```

```

times-ratio of E is 0.7
times-ratio of I is 7
e= 141.4214 SIN( 314.159 T+ 0°)
i= 13.492 SIN( 314.159 T+ -17.44059 )

```




图 1-5

§1.2 交流电路和矢量轨迹

正弦波电压和电流用矢量来处理时,可将其间的关系分别用复数阻抗和复数导纳来表示,这时,交流电路问题的求解将变得非常容易。这是因为当电路的关系式中某一个因素改变时,若想知道所对应的复数量的变化,只需看一下该复数量相应的矢量顶点所描绘的轨迹就很清楚了。我们称此轨迹为**矢量轨迹**。

1.2.1 RL 串联电路的矢量轨迹

在图 1-6 所示的 RL 串联电路上加上角频率为 ω 的电源电压时,电流 I 为:

$$\begin{aligned}
 I &= \frac{E}{R + j\omega L} \\
 &= \left(\frac{R}{R^2 + \omega^2 L^2} - j \frac{\omega L}{R^2 + \omega^2 L^2} \right) \dot{E}
 \end{aligned} \quad (1.16)$$




图 1-6 RL 串联电路

为了求得当角频率 ω 由 0 到 ∞ 变化时电流的轨迹, 可以设

$$I = x + jy$$

则

$$x = \frac{R\dot{E}}{R^2 + \omega^2 L^2}, \quad y = \frac{-\omega L\dot{E}}{R^2 + \omega^2 L^2} \quad (1.17)$$

在式(1.17)中消去变量 ω , 得到:

$$x^2 + y^2 = \frac{R^2 + \omega^2 L^2}{(R^2 + \omega^2 L^2)^2} \dot{E}^2 = \frac{\dot{E}^2}{R^2 + \omega^2 L^2} = \frac{x}{R} \dot{E} \quad (1.18)$$

可见, x 和 y 满足下面的圆方程:

$$\left(x - \frac{\dot{E}}{2R}\right)^2 + y^2 = \left(\frac{\dot{E}}{2R}\right)^2 \quad (1.19)$$

是以 $\left(\frac{\dot{E}}{2R}, 0\right)$ 为圆心坐标, $\frac{\dot{E}}{2R}$ 为半径的圆方程.

另外, 由式(1.17)可知, 在 $\omega \geq 0$ 时, $x \geq 0, y \leq 0$. 所以,

$$I = x + jy$$

的轨迹是如图 1-7 所示的半圆.




图 1-7 矢量轨迹

例题 4 (DIANLU.004) 在图 1-6 所示的 RL 串联电路上, 当电流的角频率 ω 由 0 变化到 ∞ 时, 试编程序来绘制电路中流过的电流 I 矢量轨变. 另外, 还要求每当输入一个角频率值时, 计算机就能描绘出相应的电流矢量图形.

图 1-8 表示了程序 DIANLU.004 的运行结果. 程序中: 1030 语句中的变量 E, R, L, D 分别代表电压、电阻、阻抗和图形放大倍率, 对应的数值填写在 1040 的 DATA 语句中; 1050 语句的自定义函数 $FNA(X)$ 由式(1.18)决定 $\left(y = -\sqrt{\frac{x E}{R} - x^2}\right)$, 所以只要给出 x 值就可计算出 y 值. 我们把 x, y 的坐标原点设在 CRT 监视器图象坐标的 (160, 100) 处 (即屏幕中心点) 来描绘矢量轨迹. 根据给定的 E 和 R 的数值大小, 适当调整 D 值, 使半圆的大小能在屏幕的合适位置清晰地显示出来. 从 1280 语句开始的程序段的作用是输入角频率 ω 值 (即程序中的 W), 并在其矢量轨迹上绘出电流矢量. 这是个循环次数事先不确定的循环程序段, 一旦输入 $\omega = 0$, 就结束程序的运行.

本例的输入数据为: $W = 100, 314, 159, 500, 1000, 2000$

```

1000 REM DIANLU.004 (Vector locus)
1010 TIME$ = "00 : 00 : 00" : KEY OFF
1020 'Read Voltage, Resistor, Impedance, Times-ratio
1030 READ E, R, L, D
1040 DATA 100, 10, 0.01, 22
1050 DEF FNA (X) = -SQR (E* X/R - X^ 2)
1060 SCREEN 1, 0
1070 CLS : COLOR 8, 0
1080 LINE (40, 100) - (280, 100)
1090 LINE (160, 15) - (160, 185)
1100 I = E/R
1110 H = (0 - I)/100
1120 FOR X = I TO 0 STEP H
1130 Y = FNA (X)
1140 PSET (X* D/2 + 160, -Y* D/2 + 100), 2
1150 NEXT X
1160 CIRCLE (I* D/2 + 160, FNA (I) * D/2 + 100), 1.5, 2
1170 PAINT (I* D/2 + 160, FNA (I) * D/2 + 100), 4, 2
1180 CIRCLE (160, -FNA (0) * D/2 + 100), 1.5, 2
1190 PAINT (160, -FNA (0) * D/2 + 100), 4, 2
1200 M = INT ((I* D + 160)/8)
1210 LOCATE 14, 35 : PRINT "w = 0 "
1220 LOCATE 14, 17 : PRINT "w = ∞ "
1230 LOCATE 12, 33 : PRINT I; "(A) "
1240 LOCATE 15, 1 : PRINT "E = "; E; "(V) "
1250 LOCATE 16, 1 : PRINT "R = "; R; "(Ohm) "
1260 LOCATE 17, 1 : PRINT "L = "; L; "(Henry) "
1270 LOCATE 25, 15 : PRINT "time = "; TIME$;
1280 REM --- Vector ---
1290 X0 = 2 : Y0 = 6
1300 LOCATE Y0 - 5, 5 : INPUT "W = ", W
1310 IF W = 0 THEN 1400
1320 X = R* E/(R^ 2 + W^ 2* L^ 2) : Y = FNA (X)
1330 CIRCLE (X* D/2 + 160, -Y* D/2 + 100), 1.5, 2
1340 PAINT (X* D/2 + 160, -Y* D/2 + 100), 4, 2
1350 LINE (160, 100) - (X* D/2 + 160, -Y* D/2 + 100), 1
1360 LOCATE Y0 - 5, 23
1370 PRINT USING "I=###.### +j###.### "; X; Y
1380 Y0 = Y0 + 1
1390 GOTO 1300
1400 LOCATE 22, 1 : END

```




图 1-8

1.2.2 低通滤波电路的矢量轨迹

在图 1-9 所示的低通滤波电路上, 试求 $\dot{E}_r/\dot{E}_s$ 随电源频率变化而变化的矢量轨迹. 设

$$\frac{\dot{E}_r}{\dot{E}_s} = \frac{\frac{R}{1+j\omega CR}}{j\omega L + \frac{R}{1+j\omega CR}} = x+jy \quad (1.20)$$

则

$$x = \frac{1-\delta^2}{1-\delta^2+\delta^4}, \quad y = \frac{-\delta}{1-\delta^2+\delta^4} \quad (1.21)$$